/**
 * NativeMouse Library for Adobe AIR2.*
 * net.naga.utils.NativeMouse
 * 
 * Download: http://code.google.com/p/nativemouse/downloads/list
 * Source: https://nativemouse.googlecode.com/svn/trunk/src/
 * 
 * @author		Ryo Nagai
 * @version		0.22
 */

/*
NativeMouse Library

The MIT License

Copyright (c) 2011 Ryo Nagai

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in
all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
THE SOFTWARE.
*/

package net.naga.utils
{
	import flash.desktop.NativeApplication;
	import flash.desktop.NativeProcess;
	import flash.desktop.NativeProcessStartupInfo;
	import flash.events.NativeProcessExitEvent;
	import flash.events.Event;
	import flash.events.ProgressEvent;
	import flash.system.Capabilities;
	import flash.filesystem.File;
	import flash.net.URLVariables;
	import flash.utils.IDataInput;
	import flash.geom.Point;
	
	public class NativeMouse
	{
		private static var _mousePoint:Point;
		private static var _process:NativeProcess;
		private static var _nativeProcessStartupInfo:NativeProcessStartupInfo;
		private static var _events:Array;
		
		/**
		 * NativeMouse Libraryを初期化します。
		 * <p>このメソッドは最初に一度のみ実行してください。</p>
		 */
		public static function initialize():void
		{
			_mousePoint = new Point;
			_events = new Array();
			
			// ネイティブプロセスが実行できる場合
			if (NativeProcess.isSupported)
			{
				// 「AIRMouseEventAgent」のパスを取得します。
				var file:File = File.applicationDirectory;
				file = file.resolvePath("net/naga/utils/nativeapps");
				
				// Windowsの場合
				if (Capabilities.os.toLowerCase().indexOf("win") != -1)
					file = file.resolvePath("win/AIRMouseEventAgent.exe");
				// Macの場合
				if (Capabilities.os.toLowerCase().indexOf("mac") != -1)
					file = file.resolvePath("mac/AIRMouseEventAgent");
				
				// ネイティブプロセスの設定です。
				_nativeProcessStartupInfo = new NativeProcessStartupInfo();
				_nativeProcessStartupInfo.executable = file;
				_process = new NativeProcess();
				
				// AIRアプリケーション終了時のイベントハンドラです。
				NativeApplication.nativeApplication.addEventListener(Event.EXITING, nativeApplicationExitingHandler);
			}
				// ネイティブプロセスが実行できない場合
				// AIR2.0以降のバージョンであるか、また、アプリケーションXMLファイルで
				//
				// <supportedProfiles>extendedDesktop</supportedProfiles>
				//
				// この記述があるかを確認して下さい。
			else
			{
				trace("[NativeMouse] ネイティブプロセスが実行できません。");
				trace("[NativeMouse] <supportedProfiles>extendedDesktop</supportedProfiles> を記述してください。");
			}
		}

		/**
		 * ネイティブプロセス「AIRMouseEventAgent」を開始します。
		 */
		public static　function start():void
		{
			if (NativeProcess.isSupported)
			{
				// 「AIRMouseEventAgent」を実行します。
				try
				{
					_process.start(_nativeProcessStartupInfo);
					_process.addEventListener(ProgressEvent.STANDARD_OUTPUT_DATA, updateMouse);
					_process.addEventListener(NativeProcessExitEvent.EXIT, nativeProcessExitHandler);
				}
				// 3219エラーの場合、「AIRMouseEventAgent」ファイルの実行権限がありません。
				// FlashBuilder4で net.naga.utils.nativeapps.* にある「AIRMouseEventAgent」のプロパティで
				// 実行可能にチェックをする、または、ターミナルなどからchmodコマンドでファイル属性を 744
				// に変更後、プロジェクト->クリーンを実行してください。
				catch(e:Error)
				{
					trace(e + "\n[NativeMouse] Error: 「AIRMouseEventAgent」のファイル属性（実行権限）を確認して下さい。")
				}
			}
		}
		
		/**
		 * ネイティブプロセス「AIRMouseEventAgent」を停止します。
		 */
		public static　function stop():void
		{
			_process.removeEventListener(ProgressEvent.STANDARD_OUTPUT_DATA, updateMouse);
			_process.removeEventListener(NativeProcessExitEvent.EXIT, nativeProcessExitHandler);
			_process.exit();
		}
		
		/**
		 * 「AIRMouseEventAgent」が不正終了された場合、再起動します。
		 */
		private static　function nativeProcessExitHandler(e:NativeProcessExitEvent):void
		{
			trace("[NativeMouse] 不正終了のため「AIRMouseEventAgent」を再起動します。");
			_process.removeEventListener(ProgressEvent.STANDARD_OUTPUT_DATA, updateMouse);
			_process.removeEventListener(NativeProcessExitEvent.EXIT, nativeProcessExitHandler);
			start();
		}
		
		/**
		 * マウスの各プロパティを取得し、登録されている関数を実行します。
		 */
		private static　function updateMouse(e:ProgressEvent):void
		{
			// 「AIRMouseEventAgent」からの標準出力データを取得します。
			var standardOutPut:IDataInput = _process.standardOutput;
			var rowData:String = standardOutPut.readUTFBytes(standardOutPut.bytesAvailable);
			var mousePosition:URLVariables = new URLVariables(rowData);
			
			// エラーチェックします。
			if (mousePosition.x == (mousePosition.ec - mousePosition.y))
			{
				// 各プロパティを更新します。
				_mousePoint.x = mousePosition.x;
				_mousePoint.y = mousePosition.y;
				
				// 登録されている関数を実行します。
				for each (var element:Function in _events)
				{
					element.apply(element);	
				}
			}
		}
		
		/**
		 * イベント取得の際に実行する関数を登録します。
		 * 
		 * @param event	Function型 関数
		 */
		public static function addEvent(event:Function):void
		{
			_events.push(event);
		}
		
		/**
		 * 登録した関数を削除します。
		 * 
		 * @param event	Function型 関数
		 */
		public static function dispatchEvent(event:Function):void
		{
			if (_events.indexOf(event) != -1)
				_events.splice(_events.indexOf(event));
		}

		/**
		 * マウス座標をPoint型で取得します。
		 */
		public static　function get mousePoint():Point { return _mousePoint; }
		
		/**
		 * アプリケーション終了時に「AIRMouseEventAgent」を終了します。
		 */
		private static　function nativeApplicationExitingHandler(e:Event):void
		{
			stop();
		}
	}
}